Text Reference: Section 6.4

Lab 7: Polynomial Roots via the QR-Method for Eigenvalues

The purpose of this set of exercises is to show how the real roots of a polynomial can be calculated by finding the eigenvalues of a particular matrix. These eigenvalues will be found by the QR method described below.

A **polynomial of degree** n is a function of the form

$$p(t) = a_0 + a_1 t + \ldots + a_{n-1} t^{n-1} + a_n t^n$$

where $a_0, a_1, \ldots, a_{n-1}$, and a_n are real numbers with $a_n \neq 0$. A **root** of a polynomial is a value of t for which $p(t) = 0$. It is often necessary (especially in calculus-based applications) to find all of the real roots of a given polynomial. In practice this can be a difficult problem even for a polynomial of low degree. For a polynomial of degree 2, every algebra student learns that the roots of $a t^2 + b t + c$ can be found by the quadratic formula

$$t = -\frac{b +- \sqrt{b^2 - 4 a c}}{2 a}$$

If the polynomial is of degree 3 or 4, then there are formulas somewhat resembling the quadratic formula (but much more involved) for finding all the roots of a polynomial. However there is no general formula for finding the roots of a polynomial of degree 5 or higher.

▼ Example:

Consider the monic cubic polynomial $p(t) = 6 - 5 t - 2 t^2 + t^3$ (**monic** means the leading coefficient is 1). This polynomial is factored rather easily to find that its roots are $t = 1$, $t = -2$, and $t = 3$.

Polynomial Roots using Linear Algebra

If a polynomial cannot easily be factored, numerical techniques are used to find a polynomial's roots. There are problems with this approach as well. Algorithms such as Newton's Method may not converge to a root, or may approach the root very slowly. These methods must also be applied repeatedly to find all of the roots, and usually require a cleverly chosen starting guess for the root being sought. However, there is an algorithm from linear algebra which may be used to find all real roots of a polynomial simultaneously.

The eigenvalues of an n x n matrix A are the roots of the characteristic polynomial of A, which is defined as $q(\lambda) = \det(A - \lambda I_n)$. This
polynomial of degree n, because A is n x n. So to know the eigenvalues of A is to know the roots of the monic polynomial q(λ).

To find the roots of any given monic polynomial $p(t)$, then, two problems need to be solved:

1. A way to construct a square matrix A whose characteristic polynomial $q(\lambda)$ equals $p(\lambda)$.
2. A way to find the eigenvalues of this matrix A which does not depend on finding the roots of $q(\lambda)$.

The first problem is solved by defining the **companion matrix** for a (monic) polynomial

$$p(t) = a_0 + a_1 t + ... + a_{n-1} t^{n-1} + t^n$$

Definition:

If $p(t) = a_0 + a_1 t + ... + a_{n-1} t^{n-1} + t^n$, then the **companion matrix** for p is

$$C_p = \begin{bmatrix} 0 & 1 & 0 & \cdots & 0 \\ 0 & 0 & 1 & \cdots & 0 \\ \cdots & \cdots & \cdots & \ddots & \cdots \\ 0 & 0 & 0 & \cdots & 1 \\ -a_0 & -a_1 & -a_2 & \cdots & -a_{n-1} \end{bmatrix}.$$

Example:

The companion matrix for the polynomial $p(t) = 6 - 5t - 2t^2 + t^3$ is $C_p = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ -6 & 5 & 2 \end{bmatrix}$.

Problems to be Submitted:

Problem 1.

Find the companion matrices for the following polynomials.

(a)

$p(t) = 8 - 6t + t^2$

(b)

$p(t) = 24 - 10t - 3t^2 + t^3$

(c)

$p(t) = 24 + 14t - 13t^2 - 2t^3 + t^4$

(d)

Find the characteristic polynomials of the matrices you just found in parts (a)-(c). A Maple command such as `solve(20-10*t-3*t^2+t^3=0,t)` finds the roots of each polynomial. The Maple command `CharacteristicPolynomial(M,lambda)` finds the characteristic polynomial from the matrix M.

Problem 2.

Show that the characteristic polynomial of a companion matrix for the n^{th} degree polynomial $p(t)$ is det($C_p - I_n$) = $(-1)^n p(\lambda)$ as follows.

(a)

Show that if C_p is the companion matrix for a quadratic polynomial $p(t) = a_0 + a_1 t + t^2$, then det(C_p - I_2) = $(-1)^2 p(\lambda)$ by direct computation.

(b)

Use mathematical induction to show that the result holds for n >= 2.

Hint:

Expand the necessary determinant by cofactors down the first column.

The QR Method for Eigenvalues

The companion matrix is a matrix A whose characteristic polynomial is $p(t)$. A method for finding the eigenvalues of A which does not use the characteristic polynomial is also needed. One method which accomplishes this is called the **QR method** because it is based on the QR factorization of A.

The QR factorization of an m x n matrix A requires the matrix to have linearly independent columns. Then A can be factored as A = QR , where Q is an m x n matrix with orthonormal columns and R is an n x n invertible upper triangular matrix with positive entries on its main diagonal.

Problem 3.

Suppose A is a n x n matrix. Let A = $Q_0 R_0$ be a QR factorization of A, and create $A_1 = R_0 Q_0$. Let $A_1 = Q_1 R_1$ be a QR factorization of A_1 and create $A_2 = R_1 Q_1$.

(a)

Show that A = $Q_0 A_1 Q_0^T$. (This is Exercise 23, Section 5.2.)

(b)

Show that A = ($Q_0 Q_1$) A_2 ($Q_0 Q_1$ $)^T$.

(c)

Show that $Q_0 Q_1$ is an orthogonal matrix. (This is Exercise 29, Section 6.2.)

(d)

Show that A, A_1, and A_2 all have the same eigenvalues.

The QR Method

The QR method for finding the eigenvalues of an n x n matrix A extends the process in Problem 3 to create *a sequence of matrices with the same eigenvalues*.

Step 1:

Let $A = Q_0 R_0$ be a QR factorization of A; create $A_1 = R_0 Q_0$.

Step 2:

Let $A_1 = Q_1 R_1$ be a QR factorization of A_1; create $A_2 = R_1 Q_1$.

Step m+1:

Continue this process. Once A_m has been created, then let $A_m = Q_m R_m$ be a QR factorization of A_m and create $A_{m+1} = R_m Q_m$.

Stopping Criterion:

Stop the process when the entries below the main diagonal of A_m are sufficiently small, or stop if it appears that convergence will not happen.

Example

Let A be the companion matrix for the monic cubic polynomial $p(t) = 6 - 5t - 2t^2 + t^3$; that is, A = $\begin{bmatrix} 0. & 1. & 0. \\ 0. & 0. & 1. \\ -6. & 5. & 2. \end{bmatrix}$.

The QR factorization of this matrix is

$$A = Q_0 R_0 = \begin{bmatrix} 0. & 1. & 0. \\ 0. & 0. & 1. \\ -1. & 0. & 0. \end{bmatrix} \begin{bmatrix} 6. & -5. & -2. \\ 0. & 1. & 0. \\ 0. & 0. & 1. \end{bmatrix}, \text{ so } A_1 = R_0 Q_0 = \begin{bmatrix} 2. & 6. & -5. \\ 0. & 0. & 1. \\ -1. & 0. & 0. \end{bmatrix}.$$

This operation is performed again, producing

$$A_2 = R_1 Q_1 = \begin{bmatrix} 2.236067977 & 5.366563146 & -4.472135955 \\ 0. & 2.683281573 & -2.236067977 \\ 0. & 0. & 1. \end{bmatrix}$$

$$\begin{bmatrix} 0.8944271910 & 0.4472135955 & 0. \\ 0. & 0. & 1. \\ -0.4472135955 & 0.8944271910 & 0. \end{bmatrix} = \begin{bmatrix} 4.000000000 & -3.000000000 & 5.366563146 \\ 0.9999999998 & -2.000000000 & 2.683281573 \\ -0.4472135955 & 0.8944271910 & 0. \end{bmatrix}.$$

This matrix is still far from upper triangular, so the process is continued. After 13 steps it is found that

$$A_{13} = \begin{bmatrix} 2.991249185 & 5.038287273 & 5.086237669 \\ 0.008671583910 & -1.991447448 & -1.353492217 \\ -0.000001916695814 & 0.0004401732170 & 1.000198260 \end{bmatrix},$$

so the matrix is converging to an upper triangular matrix, and its diagonal elements are converging to the roots of $p(t)=0$, which are $t = 3$, $t = -2$, and $t = 1$.

Maple Implementation of the QR Method

The following Maple procedure performs one iteration of the QR Method for a matrix A.

```
QR := proc( A::Matrix )
  local  q, r;
  q,r := LinearAlgebra[QRDecomposition](evalf(A));
  return r . q
end proc:
```

The first 13 iterates of the QR Method for the companion matrix for the monic cubic polynomial $p(t) = 6 - 5\,t - 2\,t^2 + t^3$ can be obtained with the Maple commands:

```
A0 := Matrix([[0,1,0],[0,0,1],[-6,5,2]]);
# Each iterate B will have the same eigenvalues as A0.
B:=A0:for i from 0 to 12 do
  B := QR(B);
end do;
```

Iterate 13 is the first iterate for which all entries below the diagonal are smaller than 0.01. Decisions about an intermediate result are aided by the extra function below:

```
interface(displayprecision=5);nn:=5;
F:=x->if(abs(x)<1/10^nn) then 0 else x fi;
```

then for **B**=iterate 13, **map(F,B)** will display $\begin{bmatrix} 2.99125 & -5.03829 & -5.08624 \\ -0.00867 & -1.99145 & -1.35349 \\ 0 & 0.00044 & 1.00020 \end{bmatrix}$, which has 5 display digits and numbers smaller than $\frac{1}{10^5}$ are replaced by zero. The approximate eigenvalues of **A0** are the approximate eigenvalues of **B** = iterate 13, which are the diagonal entries $\mathbf{2.99125, -1.99145, 1.00020}$. The diagonal entries are printed using Maple code **seq(B[j,j],j=1..3);**

Problem 5.

Find the approximate roots according to the QR method for the following polynomials. Compare the answers using **LinearAlgebra[Eigenvalues](A)** where A is the companion matrix for the given polynomial.

The companion matrix for polynomial $-4 + 2\,t + t^2$ can be found from maple code **LinearAlgebra[CompanionMatrix(]t^2+2*t-4)^+**, but the transpose operation can be ignored, due to determinant rule $det(C) = det(C^T)$, applied with $C = A - \lambda I$.

(a)

$$p(t) = 8 - 6\,t + t^2$$

(b)

$$p(t) = 24 + 10\,t - 3\,t^2 + t^3$$

(c)

$$p(t) = 24 + 14\,t - 13\,t^2 - 2\,t^3 + t^4$$

Problem 6.

Define $A = \begin{bmatrix} 0 & 0 & -1 & 4 & -1 & -6 \\ 0 & -2 & 2 & -5 & -2 & -5 \\ -1 & 2 & 8 & -4 & 3 & 2 \\ 4 & -5 & -4 & -6 & 1 & 0 \\ -1 & -2 & 3 & 1 & -2 & 7 \\ -6 & -5 & 2 & 0 & 7 & 10 \end{bmatrix}$

```
A := Matrix(
  [[ 0, 0,-1, 4,-1,-6],
   [ 0,-2, 2,-5,-2,-5],
   [-1, 2, 8,-4, 3, 2],
   [ 4,-5,-4,-6, 1, 0],
   [-1,-2, 3, 1,-2, 7],
   [-6,-5, 2, 0, 7,10]] );
```

Use Maple and the QR Method to make all the entries below the main diagonal of A less than 0.01. Record how many steps it takes to get this result, and then record your estimates for the eigenvalues of A.

Answer: (1) About `100` steps. (2) Same as `LinearAlgebra[Eigenvalues](A)`, to 4 digits. The basic code to use is

```
# Compute the maximum entry |A[i,j]| below the main diagonal
normF:=proc(A::Matrix) local x,i,j,n;
  n:=LinearAlgebra[RowDimension](A);
  x:=max(seq(seq(abs(A[i,j]),i=j+1..n),j=1..n-1));
RETURN (x);
end proc:

interface(displayprecision=5);nn:=2;
F:=x->if(abs(x)<1/10^nn) then 0 else x fi;

B:=A:for i from 0 to 200 do
  B := QR( B ):
  if(normF(B)<1/10^nn) then break; fi
end do;
printf("i=%d, normF(B)=%f\n",i,normF(B)); # Print iterate number
and error estimate
```

```
map(F,B); # Display B, but print zeros instead of small decimals
```

Problem 7.

Listed below are matrices C, D and E and the corresponding Maple command which creates them. For each given matrix, do enough steps of the QR method to find a matrix B with the same eigenvalues having each entry below the main diagonal of matrix B smaller than 0.1. Record the number of steps, the final result, and give estimates for the eigenvalues of each matrix.

(a)

$$C = \begin{bmatrix} 1 & -2 & 8 \\ 7 & -7 & 6 \\ 5 & 7 & -8 \end{bmatrix}$$

```
CC := Matrix([[1, -2, 8], [7, -7, 6], [5, 7, -8]]);
```

(b)

$$D = \begin{bmatrix} 4 & -2 & 3 & -7 \\ 1 & 2 & 6 & 8 \\ 8 & 5 & 1 & -5 \\ -5 & 8 & -5 & 3 \end{bmatrix}$$

```
DD := Matrix([[4, -2, 3, -7], [1, 2, 6, 8], [8, 5, 1, -5],
[-5, 8, -5, 3]]);
```

(c)

$$E = \begin{bmatrix} 2 & 6 & -3 & 4 & -9 \\ -1 & 7 & -4 & -3 & -7 \\ -6 & -6 & -1 & 6 & 5 \\ 9 & 2 & 6 & 2 & -8 \\ -7 & -8 & 6 & -9 & -1 \end{bmatrix}$$

```
EE := Matrix([[ 2,  6, -3,  4, -9],[-1,  7, -4, -3, -7],[-6,
-6, -1,  6,  5],
  [ 9,  2,  6,  2, -8],[-7, -8,  6, -9, -1]] );
```